///////////////////////////////////////////////////
// Election Administration Polarization
// Competition RD Plot
///////////////////////////////////////////////////

gl path "~/Library/CloudStorage/GoogleDrive-danmckinleythompson@gmail.com/My Drive/ElecAdminPolarization/How_Partisan_Is_Local_Election_Admin_Replication"

* Bring in the presidential election analysis data
use "$path/analysis_data/rd_election_analysis_data.dta", clear
keep if office=="pres"

* Loop over different definitions of state competitiveness and save the RD estimate using that definition
matrix B = J(20, 3, .)
qui forval c = 1/20 {
	gen competitive = abs(lag_state_vs_dem_pres-0.5)<(`c'/100) if lag_state_vs_dem_pres!=.
	rdrobust r_oos_state_year_vs_dem rv if competitive==1, vce(cluster election_id)
	matrix B[`c', 1] = e(tau_cl)
	matrix B[`c', 2] = e(tau_cl) - 1.96*e(se_tau_cl)
	matrix B[`c', 3] = e(tau_cl) + 1.96*e(se_tau_cl)	
	drop competitive
}
clear

* Plot the RD estimates across state competitiveness definitions
svmat B
rename (B1-B3) (coef upper lower)
gen cutpoint = _n/100
twoway (connected coef cutpoint, mc(gs2) lc(gs2)) ///
	(line lower cutpoint, lc(gs10) lp(dash)) ///
	(line upper cutpoint, lc(gs10) lp(dash)), ///
	xti("Competitive Cut Point") ///
	yti("Effect on Dem Pres Vote Share") yline(0) ///
	graphregion(color(white)) legend(off)
graph export "$path/output/rd_dem_pres_vs_competitive.pdf", replace
